TP 3 : corrigé

```
> restart;
```

1. Des courbes paramétrées

1.1 En cartésienne

1.1(a) Gamma1

```
> x:=t->sin(t/2):y:=t->tan(t):
> plot([x(t),y(t),t=0..4*Pi],x=-1..1,y=-6..6);
```



Points d'inflexion : avec le produit mixte

```
> pm:=t->D(x)(t)*D(D(y))(t)-D(D(x))(t)*D(y)(t):factor(pm(t
  ));
```

$$\frac{1}{4}\,(1+\tan(t)^2)\left(4\cos\left(\frac{1}{2}\,t\right)\tan(t)+\sin\left(\frac{1}{2}\,t\right)\right)$$

```
> plot(pm(t),t=0..Pi,y=-5..5);
```




```
> factor(simplify(subs(tan(t)=2*tan(t/2)/(1-tan(t/2)^2),pm
  (t))));
```

$$2\,\frac{\sin\left(\frac{1}{2}\,t\right)\left(10\cos\left(\frac{1}{2}\,t\right)^2-1\right)\left(\cos\left(\frac{1}{2}\,t\right)-1\right)^3\left(\cos\left(\frac{1}{2}\,t\right)+1\right)^3}{\cos(t)^3\,(-1+\cos(t))^3}$$

Il y a changement de signe lorsque cos(t/2) passe par 1/sqrt(10), ce qui correspond au passage de tan(t/2) par 3

```
> evalf(2*arccos(1/sqrt(10)));
```

$$2.498091544$$

```
> restart;
```

1.1(b) Gamma2

```
> x:=t->exp(t)/t:y:=t->t*exp(t):
> plot({x(t),y(t)},t=-3..3,y=-2..6);
```




```
> plot([x(t),y(t),t=-5..5],x=-5..5,y=-1..6);
```




```
> plot([x(t),y(t),t=-5..0],x=-8..0,y=-.4..0);
```




```
> plot([x(t),y(t),t=0..5],x=2..7,y=0..10);
```



Points d'inflexion : d'abord avec les variations de y'/x' :

```
> p:=t->D(y)(t)/D(x)(t):simplify(p(t));
```

$$\frac{(1+t)\,t^2}{t-1}$$

```
> factor(diff(%,t));
```

$$2\,\frac{t\,(t^2-t-1)}{(t-1)^2}$$

```
> solve(%);
```

$$0, \frac{1}{2}+\frac{1}{2}\sqrt{5}, \frac{1}{2}-\frac{1}{2}\sqrt{5}$$

```
> evalf(%);
```

0., 1.618033989, -.6180339890

Etude avec le produit mixte :

```
> pm:=t->D(x)(t)*D(D(y))(t)-D(D(x))(t)*D(y)(t):factor(pm(t
  ));
```

$$2\,\frac{(\mathbf{e}^t)^2\,(t^2-t-1)}{t^3}$$

```
> plot(pm(t),t=-3..3,pm=-3..3);
```



On retrouve les memes résultats

```
> restart;
```

1.1(c) Gamma3

```
> x:=t->t/(t^2-1):y:=t->t^2/(t-1):
> plot([x(t),y(t)],t=-3..3,y=-2..8,color=[blue,red]);
```




```
> plot([x(t),y(t),t=-5..5],x=-2..3,y=-5..6);
```



Etude du produit mixte :

```
> pm:=t->D(x)(t)*D(D(y))(t)-D(D(x))(t)*D(y)(t):factor(pm(t
  ));
```

$$-2\,\frac{t^3+3\,t+1}{(t-1)^3\,(t+1)^3}$$

```
> plot(numer(pm(t)),t=-1..2,pm=-3..3);
```



Il y a un unique point d'inflexion entre -1 et 0 : c'est conforme aux prévisions.
Recherchons le point double :

```
> solve({x(t1)=x(t2),y(t1)=y(t2)});
```

$$\{t2 = t2, t1 = t2\}, \{t2 = \mathrm{RootOf}(\_Z^2 + \_Z - 1), t1 = -1 - \mathrm{RootOf}(\_Z^2 + \_Z - 1)\}$$

```
> solve(X^2+X-1);
```

$$-\frac{1}{2}+\frac{1}{2}\sqrt{5}, -\frac{1}{2}-\frac{1}{2}\sqrt{5}$$

```
> t1,t2:=%;
```

$$t1, t2 := -\frac{1}{2}+\frac{1}{2}\sqrt{5}, -\frac{1}{2}-\frac{1}{2}\sqrt{5}$$

```
> map(simplify,[[x(t1),y(t1)],[x(t2),y(t2)]]);
```

$$[[-1, -1], [-1, -1]]$$

```
> restart;
```

1.2 En polaire

1.2(a) Gamma4

```
> rho:=t->1+sin(3*t):
> plot([rho(t),t,t=0..2*Pi],coords=polar);
```




```
> restart;
```

1.2(b) Gamma5

```
> rho:=t->2*cos(t)+sin(t):
> plot([rho(t),t,t=0..2*Pi],coords=polar);
```



Ça ressemble à une ellipse, hein ? Et bien c'est un cercle !

```
> restart;
```

1.2(c) Gamma6

```
> rho:=t->1/sqrt(t):
> plot([rho(t),t,t=0..30],x=-2..7,y=-.5..1,coords=polar);
```



```
> limit(rho(t)*sin(t),t=0);
```

$$0$$

```
> pm:=t->D(rho)(t)*2*D(rho)(t)-rho(t)*(D(D(rho))(t)-rho(t)
  );
```

$$pm := t \to 2\,\mathrm{D}(\rho)(t)^2 - \rho(t)\,(\mathrm{D}(\mathrm{D}(\rho))(t) - \rho(t))$$

```
> factor(pm(t));
```

$$\frac{1}{4}\,\frac{(2\,t-1)\,(2\,t+1)}{t^3}$$

Il y a changement d'inflexion pour t=1/2

```
> restart;
```

1.2(d) Gamma7

```
> rho:=t->ln(t):
> plot([rho(t),t,t=0..30],x=-4..4,y=-4..4,coords=polar);
```



```
> limit(rho(t)*sin(t),t=0);
```

$$0$$

```
> series(rho(t)*sin(t),t=0);
```

$$\ln(t)\,t - \frac{1}{6}\ln(t)\,t^3 + \frac{1}{120}\ln(t)\,t^5 + \mathrm{O}(t^6)$$

```
> pm:=t->D(rho)(t)*2*D(rho)(t)-rho(t)*(D(D(rho))(t)-rho(t)
  );
```

$$pm := t \to 2\,\mathrm{D}(\rho)(t)^2 - \rho(t)\,(\mathrm{D}(\mathrm{D}(\rho))(t) - \rho(t))$$

```
> factor(pm(t));
```

$$\frac{2 + \ln(t) + \ln(t)^2\,t^2}{t^2}$$

```
> num:=numer(%);
```

$$num := 2 + \ln(t) + \ln(t)^2\, t^2$$

```
> plot(num,t=0..1,y=-2..2);
```




```
> diff(num,t);
```

$$\frac{1}{t} + 2\ln(t)\,t + 2\ln(t)^2\,t$$

En se battant un peu (pas trop), on peut montrer que ce machin est >0 sur]0,1], donc le numérateur est strictement croisant, donc passe au plus une fois par zéro. La limite en 0+ et la continuité fournissent l'existence d'un point d'annulation.

```
> restart;
```

2. Des systèmes dynamiques instables

2.1 Calcul exact

```
> rsolve({u(n+2)=9/2*u(n+1)+5/2*u(n)+7,u(0)=-3,u(1)=4},u(n));
```

$$-\frac{86}{33}\left(\frac{-1}{2}\right)^n + \frac{17}{22}5^n - \frac{7}{6}$$

```
> rsolve({u(n+2)=9/2*u(n+1)+5/2*u(n)+7,u(0)=-3,u(1)=-1/4},u(n));
```

$$-\frac{11}{6}\left(\frac{-1}{2}\right)^n - \frac{7}{6}$$

```
> rsolve(u(n+2)=9/2*u(n+1)+5/2*u(n)+7,u(n));
```

$$\frac{1}{2}\left(\frac{20}{11}\mathrm{u}(0) - \frac{4}{11}\mathrm{u}(1)\right)\left(\frac{-1}{2}\right)^n - \left(-\frac{1}{11}\mathrm{u}(0) - \frac{2}{11}\mathrm{u}(1)\right)5^n + \frac{28}{33}\left(\frac{-1}{2}\right)^n + \frac{7}{22}5^n - \frac{7}{6}$$

```
> limit(%,n=infinity);
```

$$\mathrm{signum}\left(\frac{1}{11}\mathrm{u}(0) + \frac{2}{11}\mathrm{u}(1) + \frac{7}{22}\right)\infty$$

Héhé !

```
> rsolve({u(n+2)=9/2*u(n+1)+5/2*u(n)+7,u(1)=-7/4-u(0)/2},u(n));
Error, (in rsolve/single/process) more than one recurrence relation for single function
```

argl; feintons

```
> rsolve({u(n+2)=9/2*u(n+1)+5/2*u(n)+7,u(0)=u0,u(1)=-7/4-u0/2},u(n));
```

$$\frac{1}{2}\left(\frac{7}{11} + 2\,u0\right)\left(\frac{-1}{2}\right)^n + \frac{28}{33}\left(\frac{-1}{2}\right)^n - \frac{7}{6}$$

```
> limit(%,n=infinity);
```

$$\frac{-7}{6}$$

```
> restart;
```

2.2 Calcul numérique

```
> u:=n->if n=0 then -3.0 elif n=1 then 4 else
  9/2*u(n-1)+5/2*u(n-2)+7 fi;
```

u := **proc**(*n*)

option *operator*, *arrow*;

if $n = 0$ **then** -3.0 **elif** $n = 1$ **then** 4 **else** $9 / 2*\mathrm{u}(n-1) + 5 / 2*\mathrm{u}(n-2) + 7$ **end if**

end proc

```
> seq(u(k),k=0..10);
```

-3.0, 4, 17.50000000, 95.75000000, 481.6250000, 2413.687500, 12072.65625, 60368.17187, 301845.4140, $.1509231793\ 10^{7}$, $.7546163603\ 10^{7}$

```
> u(30);
```

$$.7196583520\ 10^{21}$$

C'est très long, à cause des récursions excessives.

```
> u(100);
Warning, computation interrupted
```

```
> u:=proc(n) option remember;
  if n=0 then -3.0 elif n=1 then 4 else
  9/2*u(n-1)+5/2*u(n-2)+7 fi end:
> seq(u(k),k=0..10);
```

-3.0, 4, 17.50000000, 95.75000000, 481.6250000, 2413.687500, 12072.65625, 60368.17187, 301845.4140, $.1509231793\ 10^{7}$, $.7546163603\ 10^{7}$

```
> u(30);
```

$$.7196583520\ 10^{21}$$

et c'est instantané...

```
> u(100);
```

$$.6095743360\ 10^{70}$$

idem

Changeons les conditions initiales pour avoir convergence vers -7/6

```
> u:=proc(n) option remember;
  if n=0 then -3.0 elif n=1 then -1/4 else
  9/2*u(n-1)+5/2*u(n-2)+7 fi end:
> seq(u(k),k=0..10);
```

-3.0, $\frac{-1}{4}$, -1.625000000, -.937500000, -1.281250000, -1.109375000, -1.195312500, -1.152343750, -1.173828125, -1.163085937, -1.168457028

```
> evalf(-7/6);
```

$$-1.166666667$$

Tout baigne, non ?

```
> seq(u(k),k=0..23);
```

-3.0, $\frac{-1}{4}$, -1.625000000, -.937500000, -1.281250000, -1.109375000, -1.195312500,

-1.152343750, -1.173828125, -1.163085937, -1.168457028, -1.165771468,

-1.167114176, -1.166442462, -1.166776519, -1.166600491, -1.166643508,

-1.166397014, -1.165395333, -1.160271533, -1.134710230, -1.006874867,

-.367712477, 2.828106686

```
> plot([seq([k,u(k)],k=0..23)]);
```




```
> Digits:=30:
> seq(u(k),k=0..23);
```

-3.0, $\frac{-1}{4}$, -1.625000000, -.937500000, -1.281250000, -1.109375000, -1.195312500,

-1.152343750, -1.173828125, -1.163085937, -1.168457028, -1.165771468,

-1.167114176, -1.166442462, -1.166776519, -1.166600491, -1.166643508,

-1.166397014, -1.165395333, -1.160271533, -1.134710230, -1.006874867,

-.367712477, 2.828106686

oops, avec l’option remember, il se souvient des valeurs calculées !

```
> u:=proc(n) option remember;
  if n=0 then -3.0 elif n=1 then -1/4 else
  9/2*u(n-1)+5/2*u(n-2)+7 fi end:
> seq(u(k),k=0..23);
```

-3.0, $\frac{-1}{4}$, -1.62500000000000000000000000000, -.937500000000000000000000000000,

-1.28125000000000000000000000000, -1.10937500000000000000000000000,

-1.19531250000000000000000000000, -1.15234375000000000000000000000,

-1.17382812500000000000000000000, -1.16308593750000000000000000000,

-1.16845703125000000000000000000, -1.16577148437500000000000000000,

-1.16711425781250000000000000000, -1.16644287109375000000000000000,

-1.16677856445312500000000000000, -1.16661071777343750000000000000,

-1.16669464111328125000000000000, -1.16665267944335937500000000000,

-1.16667366027832031250000000000, -1.16666316986083984375000000000,

-1.16666841506958007812500000000, -1.16666579246520996093750000000,

-1.16666710376739501953125000000, -1.16666644811630249023437500000

Ben voila !

```
> u(100);
```

$$.276767386651185370283019004178 \; 10^{21}$$

Perdu...

```
> plot([seq([k,u(k)],k=55..70)]);
```



Employons les grands moyens

```
> Digits:=100:u:=proc(n) option remember;
  if n=0 then -3.0 elif n=1 then -1/4 else
  9/2*u(n-1)+5/2*u(n-2)+7 fi end:
> k:=1:while u(k)<1 do k:=k+1 od:k;
```

$$242$$

```
> Digits:=1000:u:=proc(n) option remember;
  if n=0 then -3.0 elif n=1 then -1/4 else
  9/2*u(n-1)+5/2*u(n-2)+7 fi end:k:=1:while u(k)<1 do k:=k+1
  od:k;
```

$$2429$$

Amusant. La suite semble exploser vers n=2.4*Digits...

```
> Digits:=2000:u:=proc(n) option remember;
  if n=0 then -3.0 elif n=1 then -1/4 else
  9/2*u(n-1)+5/2*u(n-2)+7 fi end:k:=1:while u(k)<1 do k:=k+1
  od:k;
```

$$4860$$

```
> restart:
> Digits;
```

$$10$$

2.3 Un ordre de plus

```
> expand((X-3)*(X+1/3)*(X-1/2));
```

$$X^3-\frac{19}{6}X^2+\frac{1}{3}X+\frac{1}{2}$$

Ce qui répond à la question 4 !

```
> rsolve({u(n+3)=19/6*u(n+2)-1/3*u(n+1)-1/2*u(n),u(0)=1,u(1)
  =1,u(2)=1},u(n));
```

$$\frac{32}{25}\left(\frac{1}{2}\right)^n+\frac{2}{25}3^n-\frac{9}{25}\left(\frac{-1}{3}\right)^n$$

```
> rsolve({u(n+3)=19/6*u(n+2)-1/3*u(n+1)-1/2*u(n),u(0)=2,u(1)
  =4,u(2)=1},u(n));
```

$$-\frac{18}{5}\left(\frac{-1}{3}\right)^n+\frac{28}{5}\left(\frac{1}{2}\right)^n$$

```
> u:=proc(n) option remember; if n=0 then 2.0 elif n=1 then
  4 elif n=2 then 1 else 19/6*u(n-1)-1/3*u(n-2)-1/2*u(n-3)
  fi end:
```

```
> seq(u(k),k=0..30);
```

2.0, 4, 1, .833333333, .305555555, .1898148131, .0825617233, .0453960752, .02132625715, .01112026092, .00540736960, .00275345485, .001356686699, .000674671464, .000307503311, .0000705266473, -.0002165024526, -.0008628516378, -.002695459360, -.008139752868, -.02444597181, -.07335126343, -.2200604672, -.6601847391, -1.980555886, -5.941668492, -17.82500589, -53.47501789, -160.4250539, -481.2751618, -1443.825485

Meme phénomène d'instabilité numérique. Memes motifs, meme punition.

```
> rsolve(u(n+3)=19/6*u(n+2)-1/3*u(n+1)-1/2*u(n),u(n));
```

$$-\frac{1}{2}\left(-\frac{24}{25}\,\mathrm{u}(0)-\frac{64}{25}\,\mathrm{u}(1)+\frac{24}{25}\,\mathrm{u}(2)\right)\left(\frac{1}{2}\right)^n-\left(\frac{1}{50}\,\mathrm{u}(0)+\frac{1}{50}\,\mathrm{u}(1)-\frac{3}{25}\,\mathrm{u}(2)\right)3^n$$
$$+\frac{1}{3}\left(\frac{81}{50}\,\mathrm{u}(0)-\frac{189}{50}\,\mathrm{u}(1)+\frac{27}{25}\,\mathrm{u}(2)\right)\left(\frac{-1}{3}\right)^n$$

```
> rsolve({u(n+3)=19/6*u(n+2)-1/3*u(n+1)-1/2*u(n),u(0)=a,u(1)
  =b,u(2)=c},u(n));
```

$$-\frac{1}{2}\left(\frac{24}{25}\,c-\frac{64}{25}\,b-\frac{24}{25}\,a\right)\left(\frac{1}{2}\right)^n-\left(-\frac{3}{25}\,c+\frac{1}{50}\,b+\frac{1}{50}\,a\right)3^n$$
$$+\frac{1}{3}\left(\frac{27}{25}\,c-\frac{189}{50}\,b+\frac{81}{50}\,a\right)\left(\frac{-1}{3}\right)^n$$

La condition recherchée est donc : a+b-6c=0, ce qui correspond géométriquement, pour (a,b,c), à décrire un plan.